Pensieve header: One-Co computations in the abc presentation, “na nδa nca nδaa” fork; continues pensieve://2015-08/.

The bracket

On the elements β, na, c, nδa, nca, nδaa. (“n” for naive)

Generalities

```
DQ[is___] := (Sort[{is}] === Union[{is}]);
OQ[is___] := OrderedQ[{is}];

Simp[expr_] := Simplify[expr];
S[β[f_]] := β[Simp[f]];
S[na[i_, j_]] := na[i, j];
S[na[f_, i_, j_]] := na[Simp[f], i, j];
S[c[f_, k_]] := c[Simp[f], k];
S[nδa[f_, i_, j_]] := nδa[Simp[f], i, j];
S[nca[f_, j_, k_, l_]] := nca[Simp[f], j, k, l];
S[nδaa[f_, i_, j_, k_, l_]] := nδaa[Simp[f], i, j, k, l];
S[expr_] := expr /. (λ_β | λ_na | λ_nδa | λ_c | λ_nca | λ_nδaa) :> S[λ];

β[0] := 0;
β /: β[f_] + β[g_] := β[f + g] // S;
β /: g_ * β[f_] := β[g f] // S;
na[0, _, _] := 0;
na /: na[f_, j_, k_] + na[g_, j_, k_] := na[f + g, j, k] // S;
na /: g_ * na[f_, j_, k_] := na[g f, j, k] // S;
c[0, _] := 0;
c /: c[f_, j_] + c[g_, j_] := c[f + g, j] // S;
c /: g_ * c[f_, j_] := c[g f, j] // S;
nδa[0, _, _] := 0;
nδa /: nδa[f_, j_, k_] + nδa[g_, j_, k_] := nδa[f + g, j, k] // S;
nδa /: g_ * nδa[f_, j_, k_] := nδa[g f, j, k] // S;
nca[0, _, _, _] := 0;
nca /: nca[f_, j_, k_, l_] + nca[g_, j_, k_, l_] := nca[f + g, j, k, l] // S;
nca /: g_ * nca[f_, j_, k_, l_] := nca[g f, j, k, l] // S;
nδaa[0, _, _, _, _] := 0;
nδaa /: nδaa[f_, i_, j_, k_, l_] + nδaa[g_, i_, j_, k_, l_] :=
  nδaa[f + g, i, j, k, l] // S;
nδaa /: g_ * nδaa[f_, i_, j_, k_, l_] := nδaa[g f, i, j, k, l] // S;
```

The γ shortcuts

```
nγ[f_, j_, k_] := nδa[f, j, k] - c[b_j f, k] // S;
nγ[f_, j_, k_, l_] /; DQ[j, k, l] := nca[f, l, j, k] - nca[f, k, j, l] // S;
nac[f_, j_, k_, l_] := nca[f, l, j, k] + B[na[1, j, k], c[f, l]];
naop[f_, j_] := na[f, j, j] + β[-f b_j] + c[-f, j];
```

NonCommutativeMultiply

```
Unprotect[NonCommutativeMultiply];
NonCommutativeMultiply[0, _] = 0; NonCommutativeMultiply[_, 0] = 0;
NonCommutativeMultiply[x_, x_] = 0;
NonCommutativeMultiply[x_Plus, y_] := NonCommutativeMultiply[#, y] & /@ x;
NonCommutativeMultiply[x_, y_Plus] := NonCommutativeMultiply[x, #] & /@ y;

β[f_] ** na[g_, j_, k_] := na[f g, j, k];
β[f_] ** c[g_, j_] := c[f g, j];
c[g_, j_] ** β[f_] := c[f g, j];
β[f_] ** nδa[g_, j_, k_] := nδa[f g, j, k];
β[f_] ** nca[g_, i_, j_, k_] := nca[f g, i, j, k];
nca[g_, i_, j_, k_] ** β[f_] := nca[f g, i, j, k];
β[f_] ** nδaa[g_, i_, j_, k_, l_] := nδaa[f g, i, j, k, l];
nδa[g_, j_, k_] ** β[f_] := nδa[f g, j, k];
δ ** na[f_, i_, j_] := nδa[f, i, j];
c[f_, i_] ** na[g_, j_, k_] := nca[f g, i, j, k];
na[f_, i_, j_] ** nδa[g_, k_, l_] := nδaa[f g, i, j, k, l];
nδa[f_, i_, j_] ** na[g_, k_, l_] := nδaa[f g, i, j, k, l];

δ ** _c = 0;
δ ** _nδa = 0;
δ ** _nca = 0;
δ ** _nδaa = 0;
_c ** _c = 0;
_c ** _nδa = _nδa ** _c = 0;
_c ** _nca = _nca ** _c = 0;
_c ** _nδaa = _nδaa ** _c = 0;
_nδa ** _nδa = 0;
_nδa ** _nδaa = _nδaa ** _nδa = 0;
_nδa ** _nca = _nca ** _nδa = 0;
```

```
NonCommutativeMultiply::ndef =
 "NonCommuatativeMultiply is not defined on {`1`,`2`}."
NonCommutativeMultiply[x_, y_] :=
  (Message[NonCommutativeMultiply::ndef, x, y]; Undefined);
```

```
NonCommuatativeMultiply is not defined on {`1`,`2`}.
```

Bracket Generalities

```
B[0, _] = 0; B[_, 0] = 0;
B[x_, x_] = 0;
B[x_Plus, y_] := B[#, y] & /@ x;
B[x_, y_Plus] := B[x, #] & /@ y;
```

Fundamental Brackets

a-β, a-c, a-a, AS

```
B[na[j_, k_], β[g_]] := nγ[∂b_j g - ∂b_k g, j, k];
B[β[g_], na[j_, k_]] := -B[na[j, k], β[g]];
B[na[j_, k_], na[l_, m_]] /; ({j, k} ⋂ {l, m} === {}) := 0;
B[na[j_, k_], na[j_, l_]] /; DQ[j, k, l] := nγ[1, j, k, l] // S;
B[na[j_, k_], na[i_, k_]] /; DQ[i, j, k] := na[b_i, j, k] - na[b_j, i, k] // S;
B[na[j_, k_], na[k_, l_]] /; DQ[j, k, l] :=
  na[b_j, k, l] - na[b_k, j, l] - nγ[1, j, k, l] // S;
B[na[k_, l_], na[j_, k_]] /; DQ[j, k, l] := -B[na[j, k], na[k, l]];
(* backie *) B[na[j_, k_], na[k_, j_]] /; DQ[j, k] :=
  na[b_j, k, j] - na[b_k, j, k] + na[b_j, k, k] - na[b_k, j, j] + nca[1, k, k, j] -
   nca[1, j, j, k] + nca[1, k, j, j] - nca[1, j, k, k] + nγ[1, j, k] - nγ[1, k, j];
(* [tail, selfie] *) B[na[j_, k_], na[j_, j_]] /; DQ[j, k] := nγ[1, j, k] // S;
B[na[j_, j_], na[j_, k_]] /; DQ[j, k] := -B[na[j, k], na[j, j]];
(* [head, selfie] *) B[na[j_, k_], na[k_, k_]] /; DQ[j, k] := nγ[-1, j, k] // S;
B[na[k_, k_], na[j_, k_]] /; DQ[j, k] := -B[na[j, k], na[k, k]];
B[na[f_, j_, k_], c[g_, j_]] /; DQ[j, k] := nγ[-f g, j, k];
B[na[f_, j_, k_], c[g_, k_]] /; DQ[j, k] := nγ[f g, j, k];
B[na[f_, j_, k_], c[g_, l_]] /; ({j, k} ⋂ {l} === {}) := 0;
B[na[f_, j_, j_], c[g_, j_]] = 0;
B[c[g_, l_], na[f_, j_, k_]] := -B[na[f, j, k], c[g, l]];
```

Vanishing brackets

```
B[_β, _β | δ | _c | _nδa | _nca | _nδaa] = 0;
B[_β | δ | _c | _nδa | _nca | _nδaa, _β] = 0;
B[δ | _c | _nδa | _nca | _nδaa, δ | _c | _nδa | _nca | _nδaa] = 0;
```

Composite Brackets

```
B[na[f_, j_, k_], β[g_]] := β[f] ** B[na[j, k], β[g]];
B[β[g_], na[f_, j_, k_]] := -B[na[f, j, k], β[g]];
B[na[f_, j_, k_], na[l_, m_]] :=
  B[β[f], na[l, m]] ** na[1, j, k] + β[f] ** B[na[j, k], na[l, m]];
B[na[f_, j_, k_], na[g_, l_, m_]] :=
  B[na[f, j, k], β[g]] ** na[1, l, m] + β[g] ** B[na[f, j, k], na[l, m]];
B[na[f_, i_, j_], nδa[g_, k_, l_]] := δ ** B[na[f, i, j], na[g, k, l]];
B[nδa[f_, i_, j_], na[g_, k_, l_]] := δ ** B[na[f, i, j], na[g, k, l]];
B[na[f_, i_, j_], nca[g_, k_, l_, m_]] :=
  B[na[f, i, j], c[g, k]] ** na[1, l, m] + c[g, k] ** B[na[f, i, j], na[l, m]];
B[nca[g_, k_, l_, m_], na[f_, i_, j_]] := -B[na[f, i, j], nca[g, k, l, m]];
B[na[f_, i_, j_], nδaa[g_, k_, l_, m_, n_]] :=
  B[na[f, i, j], nδa[g, k, l]] ** na[1, m, n] + nδa[g, k, l] ** B[na[f, i, j], na[m, n]];
B[nδaa[g_, k_, l_, m_, n_], na[f_, i_, j_]] := -B[na[f, i, j], nδaa[g, k, l, m, n]];
```

```
B::ndef = "B is not defined on {`1`,`2`}."
B[x_, y_] := (Message[B::ndef, x, y]; Undefined);
```

```
B is not defined on {`1`,`2`}.
```

nδaa relations

```
VS = True;
```

“First sort tails then sort heads”

Standard Swinging - sorts heads, if support is 4 strands:

```
S[nδaa[f_, i_, j_, k_, l_]] /; DQ[i, j, k, l] ⋀ OQ[i, k] ⋀ ! OQ[j, l] := (
   If[VS, Print["Standard swinging on ", nδaa[f, i, j, k, l]]];
   S[nδaa[f, i, l, k, j] + nca[b_k f, l, i, j] -
     nca[b_i f, l, k, j] - nca[b_k f, j, i, l] + nca[b_i f, j, k, l]]
  );
```

Locality - sorts tails when supports are disjoint:

```
S[nδaa[f_, i_, j_, k_, l_]] /; ({i, j} ⋂ {k, l} === {}) ⋀ ! OQ[i, k] := (
   If[VS, Print["Locality on ", nδaa[f, i, j, k, l]]];
   nδaa[f, k, l, i, j] // S
  );
```

Commute Heads - sorts tails when the heads are the same:

```
S[nδaa[f_, i_, k_, j_, k_]] /; DQ[i, j, k] ⋀ ! OQ[i, j] := (
   If[VS, Print["Commute heads on ", nδaa[f, i, k, j, k]]];
   S[nδaa[f, j, k, i, k] + nδa[-bi f, j, k] + nδa[bj f, i, k]]
  );
```

Commute Head/Tail - sorts tails:

```
S[nδaa[f_, i_, j_, k_, i_]] /; DQ[i, j, k] ⋀ ! OQ[i, k] := (
   If[VS, Print["Commute head/tail on ", nδaa[f, i, j, k, i]]];
   S[
    nδaa[f, k, i, i, j] + nδaa[f, k, j, i, j] - nδaa[f, i, j, k, j]
   ]
  );
```

Commute Head/Tail - sorts heads where heads & tails are both broken:

```
S[nδaa[f_, k_, j_, j_, i_]] /; DQ[i, j, k] ⋀ OQ[i, j, k] := (
   If[VS, Print["Commute head/tail on ", nδaa[f, k, j, j, i]]];
   S[
    nδaa[f, j, i, k, j] + nδaa[f, j, i, k, i] - nδaa[f, k, i, j, i]
   ]
  );
```

2113 Swinging - sorts tails:

```
nδaa[f, j, i, ii, k] // S
```

```
Locality on nδaa[f, j, i, ii, k]

Standard swinging on nδaa[f, ii, k, j, i]

nca[-f bii, i, j, k] + nca[f bii, k, j, i] +
 nca[-f bj, k, ii, i] + nca[f bj, i, ii, k] + nδaa[f, ii, i, j, k]
```

```
S[nδaa[f_, j_, i_, i_, k_]] /; DQ[i, j, k] ⋀ OQ[i, j, k] := (
   If[VS, Print["2113 swinging on ", nδaa[f, j, i, i, k]]];
   S[nca[-f bi, i, j, k] + nca[f bi, k, j, i] +
     c[-f bj, k] ** naop[1, i] + nca[f bj, i, i, k] + naop[f, i] ** nδa[1, j, k]]
  );
```

```
nδaa[f, j, i, i, k] // S
```

```
2113 swinging on nδaa[f, j, i, i, k]

c[f bi bj, k] + nca[-f bi, i, j, k] + nca[f bi, k, j, i] +
 nca[-f bj, k, i, i] + nca[f bj, i, i, k] + nδa[-f bi, j, k] + nδaa[f, i, i, j, k]
```

3112 Swinging - sorts tails:

```
nδaa[f, k, i, ii, j] // S
```

```
Locality on nδaa[f, k, i, ii, j]

Standard swinging on nδaa[f, ii, j, k, i]

nca[-f b_ii, i, k, j] + nca[f b_ii, j, k, i] +
 nca[-f b_k, j, ii, i] + nca[f b_k, i, ii, j] + nδaa[f, ii, i, k, j]
```

```
S[nδaa[f_, k_, i_, i_, j_]] /; DQ[i, j, k] ∧ OQ[i, j, k] := (
   If[VS, Print["3112 swinging on ", nδaa[f, k, i, i, j]]];
   S[nca[-f b_i, i, k, j] + nca[f b_i, j, k, i] +
     c[-f b_k, j] ** naop[1, i] + nca[f b_k, i, i, j] + naop[1, i] ** nδa[f, k, j]]
  );
```

```
nδaa[f, k, i, i, j] // S
```

```
3112 swinging on nδaa[f, k, i, i, j]

c[f b_i b_k, j] + nca[-f b_i, i, k, j] + nca[f b_i, j, k, i] +
 nca[-f b_k, j, i, i] + nca[f b_k, i, i, j] + nδa[-f b_i, k, j] + nδaa[f, i, i, k, j]
```

Tails Commute - sorts heads when the tails are the same:

```
S[nδaa[f_, i_, j_, i_, l_]] /; DQ[i, j, l] ∧ ! OQ[j, l] := (
   If[VS, Print["Tails commute on ", nδaa[f, i, j, i, l]]];
   nδaa[f, i, l, i, j] // S
  );
```

1321 Swinging - sorts heads:

```
nδaa[f, i, k, j, ii] // S
```

```
Standard swinging on nδaa[f, i, k, j, ii]

nca[-f b_i, ii, j, k] + nca[f b_i, k, j, ii] +
 nca[-f b_j, k, i, ii] + nca[f b_j, ii, i, k] + nδaa[f, i, ii, j, k]
```

```
S[nδaa[f_, i_, k_, j_, i_]] /; DQ[i, j, k] ∧ OQ[i, j, k] := (
   If[VS, Print["1321 swinging on ", nδaa[f, i, k, j, i]]];
   S[nca[-f b_i, i, j, k] + nca[f b_i, k, j, i] +
     nca[-f b_j, k, i, i] + nac[f b_j, i, k, i] + nδaa[f, i, i, j, k]]
  );
```

1322 Swinging - sorts heads, but breaks tails:

```
nδaa[f, i, k, j, jj] // S
```

```
Standard swinging on nδaa[f, i, k, j, jj]

nca[-f b_i, jj, j, k] + nca[f b_i, k, j, jj] +
 nca[-f b_j, k, i, jj] + nca[f b_j, jj, i, k] + nδaa[f, i, jj, j, k]
```

```
S[nδaa[f_, i_, k_, j_, j_]] /; DQ[i, j, k] ⋀ OQ[i, j, k] := (
   If[VS, Print["1322 swinging on ", nδaa[f, i, k, j, j]]];
   S[nac[-f b_i, j, k, j] + nca[f b_i, k, j, j] +
     nca[-f b_j, k, i, j] + nca[f b_j, j, i, k] + nδaa[f, j, k, i, j]]
  );
```

1332 Swinging - sorts heads:

```
nδaa[f, i, k, kk, j] // S
```

```
Standard swinging on nδaa[f, i, k, kk, j]
nca[-f b_i, j, kk, k] + nca[f b_i, k, kk, j] +
 nca[-f b_kk, k, i, j] + nca[f b_kk, j, i, k] + nδaa[f, i, j, kk, k]
```

```
S[nδaa[f_, i_, k_, k_, j_]] /; DQ[i, j, k] ⋀ OQ[i, j, k] := (
   If[VS, Print["1332 swinging on ", nδaa[f, i, k, k, j]]];
   S[c[-f b_i, j] ** naop[1, k] + nca[f b_i, k, k, j] +
     nca[-f b_k, k, i, j] + nca[f b_k, j, i, k] + nδa[f, i, j] ** naop[1, k]]
  );
```

```
nδaa[f, i, k, k, j] // S
```

```
1332 swinging on nδaa[f, i, k, k, j]
c[f b_i b_k, j] + nca[-f b_i, j, k, k] + nca[f b_i, k, k, j] +
 nca[-f b_k, k, i, j] + nca[f b_k, j, i, k] + nδa[-f b_k, i, j] + nδaa[f, i, j, k, k]
```

1231 Swinging - sorts heads:

```
nδaa[f, i, j, k, ii] // S
```

```
Standard swinging on nδaa[f, i, j, k, ii]
nca[-f b_i, ii, k, j] + nca[f b_i, j, k, ii] +
 nca[-f b_k, j, i, ii] + nca[f b_k, ii, i, j] + nδaa[f, i, ii, k, j]
```

```
S[nδaa[f_, i_, j_, k_, i_]] /; DQ[i, j, k] ⋀ OQ[i, j, k] := (
   If[VS, Print["1231 swinging on ", nδaa[f, i, j, k, i]]];
   S[nac[-f b_i, k, j, i] + nca[f b_i, j, k, i] +
     nca[-f b_k, j, i, i] + nac[f b_k, i, j, i] + nδaa[f, i, i, k, j]]
  );
```

1211 sliding - sorts heads:

```
S[nδaa[f_, i_, j_, i_, i_]] /; DQ[i, j] ⋀ OQ[i, j] := (
   If[VS, Print["1211 sliding on ", nδaa[f, i, j, i, i]]];
   S[nδaa[f, i, i, i, j]]
  );
```

2111 sliding - sorts tails:

```
S[nδaa[f_, j_, i_, i_, i_]] /; DQ[i, j] ⋀ OQ[i, j] := (
   If[VS, Print["2111 sliding on ", nδaa[f, j, i, i, i]]];
   S[nδaa[f, i, i, j, i]]
  );
```

2212 sliding - sorts tails:

```
S[nδaa[f_, j_, j_, i_, j_]] /; DQ[i, j] ⋀ OQ[i, j] := (
   If[VS, Print["2212 sliding on ", nδaa[f, j, j, i, j]]];
   S[nδaa[f, i, j, j, j]]
  );
```

2221 sliding - sorts heads:

```
S[nδaa[f_, j_, j_, j_, i_]] /; DQ[i, j] ⋀ OQ[i, j] := (
   If[VS, Print["2221 sliding on ", nδaa[f, j, j, j, i]]];
   S[nδaa[f, j, i, j, j]]
  );
```

2231 Swinging - sorts heads:

```
nδaa[f, j, jj, k, i] // S
```

```
Standard swinging on nδaa[f, j, jj, k, i]
```

$$\text{nca}[-f\,b_j, i, k, jj] + \text{nca}[f\,b_j, jj, k, i] + \text{nca}[-f\,b_k, jj, j, i] + \text{nca}[f\,b_k, i, j, jj] + \text{nδaa}[f, j, i, k, jj]$$

```
S[nδaa[f_, j_, j_, k_, i_]] /; DQ[i, j, k] ⋀ OQ[i, j, k] := (
   If[VS, Print["2231 swinging on ", nδaa[f, j, j, k, i]]];
   S[nca[-f b_j, i, k, j] + nca[f b_j, j, k, i] +
     nac[-f b_k, j, i, j] + nca[f b_k, i, j, j] + nδaa[f, j, i, k, j]]
  );
```

2331 Swinging - sorts heads:

```
nδaa[f, j, k, kk, i] // S
```

```
Standard swinging on nδaa[f, j, k, kk, i]
```

$$\text{nca}[-f\,b_j, i, kk, k] + \text{nca}[f\,b_j, k, kk, i] + \text{nca}[-f\,b_{kk}, k, j, i] + \text{nca}[f\,b_{kk}, i, j, k] + \text{nδaa}[f, j, i, kk, k]$$

```
S[nδaa[f_, j_, k_, k_, i_]] /; DQ[i, j, k] ⋀ OQ[i, j, k] := (
   If[VS, Print["2331 swinging on ", nδaa[f, j, k, k, i]]];
   S[c[-f b_j, i] ** naop[1, k] + nca[f b_j, k, k, i] +
     nca[-f b_k, k, j, i] + nca[f b_k, i, j, k] + nδa[f, j, i] ** naop[1, k]]
  );
```

Backie jkkj Swinging - sorts tails or heads:

```
nδaa[f, j, k, kk, jj] // S
```

```
Standard swinging on nδaa[f, j, k, kk, jj]
```

$$\text{nca}[-f\,b_j,\ jj,\ kk,\ k] + \text{nca}[f\,b_j,\ k,\ kk,\ jj] + \text{nca}[-f\,b_{kk},\ k,\ j,\ jj] + \text{nca}[f\,b_{kk},\ jj,\ j,\ k] + \text{nδaa}[f,\ j,\ jj,\ kk,\ k]$$

```
S[nδaa[f_, j_, k_, k_, j_]] /; DQ[j, k] := (
   If[VS, Print["Backie swinging on ", nδaa[f, j, k, k, j]]];
   S[c[-f b_j, j] ** naop[1, k] + nca[f b_j, k, k, j] +
     nca[-f b_k, k, j, j] + nac[f b_k, j, k, j] + nδa[f, j, j] ** naop[1, k]]
  );
```

Naive / EK Conversions

```
ToNaive[expr_] := expr /. {
   a[f_, j_, k_] /; DQ[j, k] :> na[f, j, k],
   a[f_, j_, j_] :>
  }
```

Testing Jacobi and Anti-Symmetry

```
FormalBasis[S_List, f_] := Module[{ff, n = Length@S, i, j, k, l},
   ff = f @@ Table[b_S[[i]], {i, n}];
   Flatten@{
      β[ff],
      Table[na[ff, S[[i]], S[[j]]], {i, n}, {j, n}],
      Table[c[ff, S[[i]]], {i, n}],
      Table[nδa[ff, S[[i]], S[[j]]], {i, n}, {j, n}],
      Table[nca[ff, S[[i]], S[[j]], S[[k]]], {i, n}, {j, n}, {k, n}],
      Table[nδaa[ff, S[[i]], S[[j]], S[[k]], S[[l]]], {i, n}, {j, n}, {k, i, n}, {l, j, n}]
     } /. 1[___] → 1
  ];
FormalBasis[n_Integer, f_] := FormalBasis[Range[n], f];
FormalPlusBasis[n_, f_] := Module[{ff},
   ff = f @@ Table[b_i, {i, n}];
   Flatten@{
      β[ff],
      Table[na[ff, i, j], {i, n - 1}, {j, i + 1, n}],
      Table[c[ff, i], {i, n}],
      Table[nδa[ff, i, j], {i, n - 1}, {j, i + 1, n}],
      Table[nca[ff, i, j, k], {i, n}, {j, n - 1}, {k, j + 1, n}],
      Table[nδaa[ff, i, j, k, l], {i, n - 1}, {j, i + 1, n}, {k, n - 1}, {l, k + 1, n}]
     } /. 1[___] → 1
  ];

VS = False;
AS[x1_, x2_] := Module[{as},
   as = B[x1, x2] + B[x2, x1] // S;
   If[as === 0, as, {x1, x2} → as]
  ];
DeleteCases[Flatten[Outer[
   AS,
   FormalPlusBasis[3, f],
   FormalPlusBasis[3, g]
  ]], 0]
```

```
{}
```

```
AS[x1_, x2_] := Module[{as},
   as = B[x1, x2] + B[x2, x1] // S;
   If[as === 0, as, {x1, x2} → as]
  ];
DeleteCases[Flatten[Outer[
   AS,
   FormalBasis[3, f],
   FormalBasis[3, g]
  ]], 0]
```

```
{}
```

```
Jacobi[x1_, x2_, x3_] := Module[{Jac},
   Jac = S[B[x1, B[x2, x3]] + B[x2, B[x3, x1]] + B[x3, B[x1, x2]]];
   If[Jac === 0, Jac, {x1, x2, x3} → Jac]
  ];
JacPlusErrors = DeleteCases[
  bas1 = FormalPlusBasis[4, f];
  bas2 = FormalPlusBasis[4, g];
  bas3 = FormalPlusBasis[4, h];
  Flatten[
   Table[Jacobi[bas1[[i]], bas2[[j]], bas3[[k]]],
    {i, Length[bas1] - 1}, {j, i + 1, Length@bas2}, {k, i + 1, Length@bas3}]
  ],
  0]
```

```
{}
```

```
VS = False;
JacErrors = DeleteCases[
  bas1 = FormalBasis[4, f];
  bas2 = FormalBasis[4, g];
  bas3 = FormalBasis[4, h];
  Flatten[
   Table[Jacobi[bas1[[i]], bas2[[j]], bas3[[k]]],
    {i, Length[bas1] - 1}, {j, i + 1, Length@bas2}, {k, i + 1, Length@bas3}]
  ],
  0]
```

```
{}
```

The Adjoint action

AutoAd

```
AutoAd[x_][y_] :=
  Module[{pows, states, i, s, seq, sh = 5, dseq, sf1, sf2, sf, t1, n},
   pows = NestList[B[x, #] &, y, 20];
   Print["pows computed for ", {x, y}, "..."];
   states = Union[Cases[pows,
      s_β | s_na | s_c | s_nδa | s_nca | s_nδaa :> ReplacePart[s, 1 → _], ∞]];
   Sum[
    seq = Cases[{#}, states[[i]], ∞] & /@ pows;
    seq = Replace[seq, {{_[f_, ___]} :> f, {} → 0}, {1}];
    Print["seq computed... ", states[[i]], " is ", i, "/", Length@states];
    dseq = Drop[seq, sh];
    If[Union[Length[MonomialList[#]] & /@ dseq] === {1} ∧
      Union[Length[FactorTermsList[#]] & /@ dseq] === {2},
     sf1 = FindSequenceFunction[FactorTermsList[#][[1]] & /@ dseq];
     sf2 = FindSequenceFunction[FactorTermsList[#][[2]] & /@ dseq];
     Print["sf1: ", sf1, " sf2: ", sf2];
     sf = (sf1[#] sf2[#] &),
     (*Else*) sf = FindSequenceFunction[dseq,
       FunctionSpace → {"ConstantRecursive", "HolonomicSequence",
         "Polynomial", "RationalFunction", "HypergeometricTerm"}];
     Print["sf: ", sf];
    ];
    ReplacePart[states[[i]], 1 → Simplify[∑_{n=0}^{sh-1} seq[[n+1]]/n! + ∑_{n=sh}^{∞} sf[n+1-sh]/n!]],
    {i, Length@states}]
  ];
(* Hint: Perhaps improve using Variables, CoefficientList, FromCoefficientList *)
```

```
AutoAd[na[t, j, k]][na[1, k, j]]
```

```
pows computed for {na[t, j, k], na[1, k, j]}...
seq computed... c[_, j] is 1/20
sf1: 7 + 2 #1 & sf2: t^4 b_j^3 (t b_j)^#1 b_k &
seq computed... c[_, k] is 2/20
sf: t^4 b_j^3 (b_j (-t b_j)^#1 + (-t b_j)^#1 b_k + 7 (t b_j)^#1 b_k + 2 #1 (t b_j)^#1 b_k) &
seq computed... na[_, j, j] is 3/20
sf1: -1 & sf2: t^4 b_j^3 (t b_j)^#1 b_k &
```

seq computed... na[_, j, k] is 4/20

sf1: -1 & sf2: $t^4 b_j^3 (t b_j)^{\#1} b_k$ &

seq computed... na[_, k, j] is 5/20

sf1: 1 & sf2: $t^4 b_j^4 (t b_j)^{\#1}$ &

seq computed... na[_, k, k] is 6/20

sf1: 1 & sf2: $t^4 b_j^4 (t b_j)^{\#1}$ &

seq computed... nca[_, j, j, k] is 7/20

sf: $-t^4 b_j^2 (t b_j)^{\#1} (b_j - 6 b_k - 2 \#1 b_k)$ &

seq computed... nca[_, j, k, k] is 8/20

sf1: -7 - 2 #1 & sf2: $t^4 b_j^3 (t b_j)^{\#1}$ &

seq computed... nca[_, k, j, j] is 9/20

sf: $-t^4 b_j^2 \left(b_j (-t b_j)^{\#1} + (-t b_j)^{\#1} b_k + 4 (t b_j)^{\#1} b_k + \#1 (t b_j)^{\#1} b_k\right)$ &

seq computed... nca[_, k, j, k] is 10/20

sf: $-t^4 b_j^2 \left(b_j (-t b_j)^{\#1} + b_j (t b_j)^{\#1} + (-t b_j)^{\#1} b_k - 2 (t b_j)^{\#1} b_k - \#1 (t b_j)^{\#1} b_k\right)$ &

seq computed... nca[_, k, k, j] is 11/20

sf1: 4 + #1 & sf2: $t^4 b_j^3 (t b_j)^{\#1}$ &

seq computed... nca[_, k, k, k] is 12/20

sf1: -3 - #1 & sf2: $t^4 b_j^3 (t b_j)^{\#1}$ &

seq computed... nδa[_, j, j] is 13/20

sf1: -3 - #1 & sf2: $t^4 b_j^2 (t b_j)^{\#1} b_k$ &

seq computed... nδa[_, j, k] is 14/20

sf: $-t^4 b_j^2 \left(b_j (-t b_j)^{\#1} + (-t b_j)^{\#1} b_k + 4 (t b_j)^{\#1} b_k + \#1 (t b_j)^{\#1} b_k\right)$ &

seq computed... nδa[_, k, j] is 15/20

sf1: -4 - #1 & sf2: $t^4 b_j^3 (t b_j)^{\#1}$ &

seq computed... nδa[_, k, k] is 16/20

sf1: -3 - #1 & sf2: $t^4 b_j^3 (t b_j)^{\#1}$ &

seq computed... nδaa[_, j, j, j, k] is 17/20

sf: $t^4 b_j \left(b_j (-t b_j)^{\#1} + b_j (t b_j)^{\#1} + (-t b_j)^{\#1} b_k - 5 (t b_j)^{\#1} b_k - 2 \#1 (t b_j)^{\#1} b_k\right)$ &

seq computed... nδaa[_, j, j, k, k] is 18/20

sf1: 2 (3 + #1) & sf2: $t^4 b_j^2 (t b_j)^{\#1}$ &

seq computed... nδaa[_, j, k, j, k] is 19/20

sf: $t^4 b_j \left(b_j (-t b_j)^{\#1} + b_j (t b_j)^{\#1} + (-t b_j)^{\#1} b_k - 5 (t b_j)^{\#1} b_k - 2 \#1 (t b_j)^{\#1} b_k\right)$ &

seq computed... nδaa[_, j, k, k, k] is 20/20

sf1: 2 (3 + #1) & sf2: $t^4 b_j^2 (t b_j)^{\#1}$ &

$$c\left[\frac{\left(1-e^{t b_j}+2 e^{t b_j} t b_j\right) b_k}{b_j}, j\right]+c\left[-1+e^{-t b_j}+\frac{e^{-t b_j}\left(1-e^{2 t b_j}+2 e^{2 t b_j} t b_j\right) b_k}{b_j}, k\right]+$$
$$na\left[e^{t b_j}, k, j\right]+na\left[-1+e^{t b_j}, k, k\right]+na\left[-\frac{\left(-1+e^{t b_j}\right) b_k}{b_j}, j, j\right]+$$
$$na\left[-\frac{\left(-1+e^{t b_j}\right) b_k}{b_j}, j, k\right]+nca\left[e^{t b_j} t, k, k, j\right]+$$
$$nca\left[\frac{-1+e^{t b_j}-2 e^{t b_j} t b_j}{b_j}, j, k, k\right]+nca\left[\frac{-1+e^{t b_j}-e^{t b_j} t b_j}{b_j}, k, k, k\right]+$$
$$nca\left[\frac{-2\left(-1+e^{t b_j}\right) b_k+b_j\left(1-e^{t b_j}+2 e^{t b_j} t b_k\right)}{b_j^2}, j, j, k\right]+$$
$$nca\left[\frac{e^{-t b_j}\left(\left(-1+e^{t b_j}\right) b_k+b_j\left(-1+e^{t b_j}-e^{2 t b_j} t b_k\right)\right)}{b_j^2}, k, j, j\right]+$$
$$nca\left[\frac{1}{b_j^2} e^{-t b_j}\left(-\left(1-3 e^{t b_j}+2 e^{2 t b_j}\right) b_k+b_j\left(-\left(-1+e^{t b_j}\right)^2+e^{2 t b_j} t b_k\right)\right), k, j, k\right]+$$
$$n\delta a\left[-e^{t b_j} t, k, j\right]+n\delta a\left[\frac{-1+e^{t b_j}-e^{t b_j} t b_j}{b_j}, k, k\right]+n\delta a\left[-\frac{\left(1-e^{t b_j}+e^{t b_j} t b_j\right) b_k}{b_j^2}, j, j\right]+$$
$$n\delta a\left[\frac{e^{-t b_j}\left(\left(-1+e^{t b_j}\right) b_k+b_j\left(-1+e^{t b_j}-e^{2 t b_j} t b_k\right)\right)}{b_j^2}, j, k\right]+$$
$$n\delta aa\left[\frac{2-2 e^{t b_j}+2 e^{t b_j} t b_j}{b_j^2}, j, j, k, k\right]+n\delta aa\left[\frac{2-2 e^{t b_j}+2 e^{t b_j} t b_j}{b_j^2}, j, k, k, k\right]+$$
$$n\delta aa\left[\frac{1}{b_j^3} e^{-t b_j}\left(\left(1-4 e^{t b_j}+3 e^{2 t b_j}\right) b_k+b_j\left(\left(-1+e^{t b_j}\right)^2-2 e^{2 t b_j} t b_k\right)\right), j, j, j, k\right]+$$
$$n\delta aa\left[\frac{1}{b_j^3} e^{-t b_j}\left(\left(1-4 e^{t b_j}+3 e^{2 t b_j}\right) b_k+b_j\left(\left(-1+e^{t b_j}\right)^2-2 e^{2 t b_j} t b_k\right)\right), j, k, j, k\right]$$

Ad

```
Ad[na[t_, j_, k_]][β[f_]] /; FreeQ[t, b_] :=
  β[f] + c[(1 - e^(-t b_j)) (∂_(b_k) f - ∂_(b_j) f), k] + nδa[((e^(-t b_j) - 1) (∂_(b_k) f - ∂_(b_j) f))/b_j, j, k];
Ad[na[t_, j_, k_]][na[1, j_, k_]] /; FreeQ[t, b_] := na[1, j, k];
Ad[na[t_, j_, k_]][na[1, n_, i_]] /;
   FreeQ[t, b_] ∧ ({j, k} ⋂ {n, i} === {}) := na[1, n, i];
Ad[na[t_, j_, k_]][na[1, j_, j_]] /; DQ[j, k] ∧ FreeQ[t, b_] :=
  na[1, j, j] + c[-1 + e^(-t b_j), k] + nδa[(1 - e^(-t b_j))/b_j, j, k];
Ad[na[t_, j_, k_]][na[1, k_, k_]] /; DQ[j, k] ∧ FreeQ[t, b_] :=
  na[1, k, k] + c[1 - e^(-t b_j), k] + nδa[(-1 + e^(-t b_j))/b_j, j, k];
Ad[na[t_, j_, k_]][na[1, i_, j_]] /; DQ[i, j, k] ∧ FreeQ[t, b_] :=
```

```
na[1, i, j] + na[1 - e^(-t b_j), i, k] + na[((e^(-t b_j) - 1) b_i)/b_j, j, k] + nca[(1 - e^(-t b_j))/b_j, k, i, j] +
 nca[(e^(-t b_j) - 1)/b_j, j, i, k] + nca[(b_i (1 - e^(-t b_j) - t b_j))/b_j^2, j, j, k] +
 nca[(e^(-2 t b_j) b_i (1 - e^(t b_j) - e^(t b_j) (e^(t b_j) - 2) t b_j))/b_j^2, k, j, k] + nca[(e^(-2 t b_j) (e^(t b_j) (1 - t b_j) - 1))/b_j,
  k, i, k] + nδa[(b_i (1 - e^(-t b_j) - t b_j))/b_j^2 + (-b_i (1 - e^(-2 t b_j) + (-1 - e^(-t b_j)) t b_j))/b_j^2, j, k] +
 nδa[(-1 + e^(-t b_j) + t b_j)/b_j + (1 - e^(-2 t b_j) + (-1 - e^(-t b_j)) t b_j)/b_j, i, k] +
 nδaa[(2 e^(-t b_j) b_i (Sinh[t b_j] - t b_j))/b_j^3, j, k, j, k] + nδaa[(-1 + e^(-t b_j) + t b_j)/b_j^2, i, j, j, k] +
 nδaa[-(1 - e^(-2 t b_j) + (-1 - e^(-t b_j)) t b_j)/b_j^2, i, k, j, k];
Ad[na[t_, j_, k_]][na[1, i_, k_]] /; DQ[i, j, k] ∧ FreeQ[t, b_] :=
 na[e^(-t b_j), i, k] + na[((1 - e^(-t b_j)) b_i)/b_j, j, k] + nca[(2 e^(-t b_j) b_i (Sinh[t b_j] - t b_j))/b_j^2, k, j, k] +
 nca[(e^(-2 t b_j) (1 + e^(t b_j) (-1 + t b_j)))/b_j, k, i, k] + nδa[(e^(-2 t b_j) b_i (-1 + e^(t b_j) (1 - t b_j)))/b_j^2, j, k] +
 nδa[(e^(-2 t b_j) (1 - e^(t b_j) (1 - t b_j)))/b_j, i, k] + nδaa[(2 e^(-t b_j) b_i (-Sinh[t b_j] + t b_j))/b_j^3, j, k, j, k] +
 nδaa[(e^(-2 t b_j) (-1 + e^(t b_j) (1 - t b_j)))/b_j^2, i, k, j, k];
Ad[na[t_, j_, k_]][na[1, j_, l_]] /; DQ[j, k, l] ∧ FreeQ[t, b_] :=
 na[1, j, l] + nca[t, l, j, k] + nca[(e^(-t b_j) - 1)/b_j, k, j, l] + nδaa[(1 - e^(-t b_j) - t b_j)/b_j^2, j, k, j, l];
Ad[na[t_, j_, k_]][na[1, k_, l_]] /; DQ[j, k, l] ∧ FreeQ[t, b_] :=
 na[e^(t b_j), k, l] + na[-((-1 + e^(t b_j)) b_k)/b_j, j, l] + nca[(-1 + e^(t b_j) (1 - t b_j))/b_j, k, k, l] +
 nca[(b_j - e^(-t b_j) b_j + b_k + e^(t b_j) (-1 + t b_j) b_k)/b_j^2, k, j, l] +
 nca[(b_j + b_k + t b_j b_k - e^(t b_j) (b_j + b_k))/b_j^2, l, j, k] + nδaa[(1 + e^(t b_j) (-1 + t b_j))/b_j^2, j, k, k, l] +
 nδaa[1/b_j^3 e^(-t b_j) (b_j + e^(2 t b_j) (b_j + (2 - t b_j) b_k) - e^(t b_j) (2 b_k + b_j (2 + t b_k))), j, k, j, l];
Ad[na[t_, j_, k_]][na[1, k_, j_]] /; DQ[j, k] ∧ FreeQ[t, b_] :=
 na[e^(t b_j), k, j] + na[-1 + e^(t b_j), k, k] + na[-((-1 + e^(t b_j)) b_k)/b_j, j, j] +
```

```
    na[-((-1 + E^(t b[j])) b[k])/b[j], j, k] + c[((1 - E^(t b[j]) + 2 E^(t b[j]) t b[j]) b[k])/b[j], j] +
    c[-1 + E^(-t b[j]) + (E^(-t b[j]) (1 - E^(2 t b[j]) + 2 E^(2 t b[j]) t b[j]) b[k])/b[j], k] + nca[E^(t b[j]) t, k, k, j] +
    nca[(-1 + E^(t b[j]) - 2 E^(t b[j]) t b[j])/b[j], j, k, k] + nca[(-1 + E^(t b[j]) - E^(t b[j]) t b[j])/b[j], k, k, k] +
    nca[(-2 (-1 + E^(t b[j])) b[k] + b[j] (1 - E^(t b[j]) + 2 E^(t b[j]) t b[k]))/b[j]^2, j, j, k] +
    nca[(E^(-t b[j]) ((-1 + E^(t b[j])) b[k] + b[j] (-1 + E^(t b[j]) - E^(2 t b[j]) t b[k])))/b[j]^2, k, j, j] +
    nca[1/b[j]^2 E^(-t b[j]) (-(1 - 3 E^(t b[j]) + 2 E^(2 t b[j])) b[k] + b[j] (-(-1 + E^(t b[j]))^2 + E^(2 t b[j]) t b[k])), k, j, k] +
    nδa[-E^(t b[j]) t, k, j] + nδa[(-1 + E^(t b[j]) - E^(t b[j]) t b[j])/b[j], k, k] + nδa[-((1 - E^(t b[j]) + E^(t b[j]) t b[j]) b[k])/b[j]^2, j, j] +
    nδa[(E^(-t b[j]) ((-1 + E^(t b[j])) b[k] + b[j] (-1 + E^(t b[j]) - E^(2 t b[j]) t b[k])))/b[j]^2, j, k] +
    nδaa[(2 - 2 E^(t b[j]) + 2 E^(t b[j]) t b[j])/b[j]^2, j, j, k, k] + nδaa[(2 - 2 E^(t b[j]) + 2 E^(t b[j]) t b[j])/b[j]^2, j, k, k, k] +
    nδaa[1/b[j]^3 E^(-t b[j]) ((1 - 4 E^(t b[j]) + 3 E^(2 t b[j])) b[k] + b[j] ((-1 + E^(t b[j]))^2 - 2 E^(2 t b[j]) t b[k])), j, j, j, k] +
    nδaa[1/b[j]^3 E^(-t b[j]) ((1 - 4 E^(t b[j]) + 3 E^(2 t b[j])) b[k] + b[j] ((-1 + E^(t b[j]))^2 - 2 E^(2 t b[j]) t b[k])), j, k, j, k];
Ad[na[t_, j_, k_]][c[1, i_]] /; FreeQ[t, b_] ∧ ({j, k} ⋂ {i} === {}) := c[1, i];
Ad[na[t_, j_, k_]][c[1, j_]] /; DQ[j, k] ∧ FreeQ[t, b_] :=
  c[1, j] + c[1 - E^(-t b[j]), k] + nδa[(E^(-t b[j]) - 1)/b[j], j, k];
Ad[na[t_, j_, k_]][c[1, k_]] /; DQ[j, k] ∧ FreeQ[t, b_] :=
  c[E^(-t b[j]), k] + nδa[(1 - E^(-t b[j]))/b[j], j, k];

Ad[x_β | x_c | x_nδa | x_nca | x_nδaa][y_] := y + B[x, y];

Ad[x_][na[f_, i_, j_]] /; f =!= 1 := Ad[x][β[f]] ** Ad[x][na[1, i, j]];
Ad[x_][c[f_, i_]] /; f =!= 1 := Ad[x][β[f]] ** Ad[x][c[1, i]];
Ad[x_][nδa[f_, j_, k_]] := δ ** (β[f] ** Ad[x][na[1, j, k]]);
Ad[x_][nca[f_, i_, j_, k_]] := Ad[x][c[f, i]] ** Ad[x][na[1, j, k]];
Ad[x_][nδaa[f_, i_, j_, k_, l_]] := Ad[x][nδa[f, i, j]] ** Ad[x][na[1, k, l]];
Ad[x_][y_Plus] := Ad[x] /@ y;

Ad::ndef = "Ad[`1` is not defined on `2`.";
Ad[x_][y_] := (Message[Ad::ndef, x, y]; Undefined);
```

AutoAd - Ad tests

```
Module[{t1, t2},
    {t1 = S[AutoAd[na[t, j, k]][#]],
     S[Ad[na[t, j, k]][#] - t1]}
  ] & @ na[1, i, k]
```

```
pows computed for {na[t, j, k], na[1, i, k]}...
```

```
seq computed... na[_, i, k] is 1/9
```

```
sf1: (-1)^#1 & sf2: t^4 b_j^4 (t b_j)^#1 &
```

```
seq computed... na[_, j, k] is 2/9
```

```
sf1: (-1)^(1+#1) & sf2: t^4 b_i b_j^3 (t b_j)^#1 &
```

```
seq computed... nca[_, k, i, k] is 3/9
```

```
sf1: (-1)^#1 (-5 + 2^(4+#1) - #1) & sf2: t^4 b_j^3 (t b_j)^#1 &
```

```
seq computed... nca[_, k, j, k] is 4/9
```

```
sf1: -2 (-1)^#1 (-4 + 2^(3+#1) - #1) & sf2: t^4 b_i b_j^2 (t b_j)^#1 &
```

```
seq computed... nδa[_, i, k] is 5/9
```

$$\text{sf1: } \frac{1}{9}\,2^{-1-\#1}\left(-81\,(-2)^{\#1}-63\,(-2)^{\#1}\,(-1)^{\#1}-29\times 2^{\#1}+29\,(-1)^{2\,\#1}\,2^{\#1}+9\times 2^{3+2\,\#1}+27\,(-1)^{\#1}\,2^{3+2\,\#1}-3\times 2^{1+\#1}\,\#1-9\,(-1)^{\#1}\,2^{1+\#1}\,\#1-3\,(-1)^{2\,\#1}\,2^{2+\#1}\,\#1\right)\ \&\ \text{sf2: } t^4\,b_j^3\,(t\,b_j)^{\#1}\ \&$$

```
seq computed... nδa[_, j, k] is 6/9
```

$$\text{sf1: } -\frac{1}{9}\,2^{-1-\#1}\left(-81\,(-2)^{\#1}-63\,(-2)^{\#1}\,(-1)^{\#1}-29\times 2^{\#1}+29\,(-1)^{2\,\#1}\,2^{\#1}+9\times 2^{3+2\,\#1}+27\,(-1)^{\#1}\,2^{3+2\,\#1}-3\times 2^{1+\#1}\,\#1-9\,(-1)^{\#1}\,2^{1+\#1}\,\#1-3\,(-1)^{2\,\#1}\,2^{2+\#1}\,\#1\right)\ \&\ \text{sf2: } t^4\,b_i\,b_j^2\,(t\,b_j)^{\#1}\ \&$$

```
seq computed... nδaa[_, i, k, j, k] is 7/9
```

$$\text{sf1: } -\frac{1}{9}\,2^{-1-\#1}\left(-81\,(-2)^{\#1}-63\,(-2)^{\#1}\,(-1)^{\#1}-29\times 2^{\#1}+29\,(-1)^{2\,\#1}\,2^{\#1}+9\times 2^{3+2\,\#1}+27\,(-1)^{\#1}\,2^{3+2\,\#1}-3\times 2^{1+\#1}\,\#1-9\,(-1)^{\#1}\,2^{1+\#1}\,\#1-3\,(-1)^{2\,\#1}\,2^{2+\#1}\,\#1\right)\ \&\ \text{sf2: } t^4\,b_j^2\,(t\,b_j)^{\#1}\ \&$$

```
seq computed... nδaa[_, j, k, i, k] is 8/9
```

$$\text{sf1: } -\frac{1}{3}\,2^{-1-\#1}\left(-3\,(-2)^{\#1}-9\,(-2)^{\#1}\,(-1)^{\#1}+13\times 2^{\#1}+17\,(-1)^{2\,\#1}\,2^{\#1}-3\times 2^{3+2\,\#1}+3\,(-1)^{\#1}\,2^{3+2\,\#1}-3\,(-2)^{\#1}\,(-1)^{\#1}\,\#1+3\times 2^{\#1}\,\#1+3\,(-1)^{2\,\#1}\,2^{1+\#1}\,\#1\right)\ \&\ \text{sf2: } t^4\,b_j^2\,(t\,b_j)^{\#1}\ \&$$

```
seq computed... nδaa[_, j, k, j, k] is 9/9
```

```
sf1: 2 (-1)^#1 (-4 + 2^(3+#1) - #1) & sf2: t^4 b_i b_j (t b_j)^#1 &
```

$$\Big\{na\Big[e^{-t\,b_j},\,i,\,k\Big]+na\Big[\frac{\left(1-e^{-t\,b_j}\right)b_i}{b_j},\,j,\,k\Big]+$$
$$nca\Big[\frac{e^{-2\,t\,b_j}\,b_i\,\left(-1+e^{2\,t\,b_j}-2\,e^{t\,b_j}\,t\,b_j\right)}{b_j^2},\,k,\,j,\,k\Big]+nca\Big[\frac{e^{-2\,t\,b_j}\,\left(1-e^{t\,b_j}+e^{t\,b_j}\,t\,b_j\right)}{b_j},\,k,\,i,\,k\Big]+$$
$$n\delta a\Big[-\frac{e^{-2\,t\,b_j}\,b_i\,\left(1-e^{t\,b_j}+e^{t\,b_j}\,t\,b_j\right)}{b_j^2},\,j,\,k\Big]+n\delta a\Big[\frac{e^{-2\,t\,b_j}\,\left(1-e^{t\,b_j}+e^{t\,b_j}\,t\,b_j\right)}{b_j},\,i,\,k\Big]+$$
$$n\delta aa\Big[-\frac{e^{-2\,t\,b_j}\,b_i\,\left(-1+e^{2\,t\,b_j}-2\,e^{t\,b_j}\,t\,b_j\right)}{b_j^3},\,j,\,k,\,j,\,k\Big]+$$
$$n\delta aa\Big[\frac{e^{-2\,t\,b_j}\,\left(-1+e^{t\,b_j}-e^{t\,b_j}\,t\,b_j\right)}{b_j^2},\,i,\,k,\,j,\,k\Big],\,0\Big\}$$

```
AdTests[na[t, j, k]] =
  {β[f[b_j, b_k]], na[1, j, k], na[1, n, i], na[1, j, j], na[1, k, k], c[1, i], c[1, j],
   c[1, k], na[1, j, l], na[1, i, j], na[1, i, k], na[1, k, l], na[1, k, j]};

S[AutoAd[na[t, j, k]][#] - Ad[na[t, j, k]][#]] & /@ Take[AdTests[na[t, j, k]], All]
```

```
$Aborted
```

The semi group properties

```
Module[{t1, t2, t3, t4},
    t1 = Ad[na[t, j, k]][#] /. (h : (β | na | c | nδa | nca | nδaa))[c_, r___] :>
       h[SeriesCoefficient[c, {t, 0, 1}], r];
    t2 = B[na[1, j, k], #];
    t3 = # // Ad[na[t, j, k]] // Ad[na[s, j, k]];
    t4 = # // Ad[na[t + s, j, k]];
    # → S[{t1 == t2, t3 - t4}]
   ] & /@ AdTests[na[t, j, k]] // ColumnForm
```

```
β[f[b_j, b_k]] → {True, 0}
na[1, j, k] → {True, 0}
na[1, n, i] → {True, 0}
na[1, j, j] → {True, 0}
na[1, k, k] → {True, 0}
c[1, i] → {True, 0}
c[1, j] → {True, 0}
c[1, k] → {True, 0}
na[1, j, l] → {True, 0}
na[1, i, j] → {True, 0}
na[1, i, k] → {True, 0}
na[1, k, l] → {True, 0}
na[1, k, j] → {True, 0}
```

Verifying R3

```
VerifyR3[t_, expr_] := Module[{lhs, rhs},
   lhs = expr // R[t, 1, 2] // R[t, 1, 3] // R[t, 2, 3] // S;
   rhs = expr // R[t, 2, 3] // R[t, 1, 3] // R[t, 1, 2] // S;
   expr → S[lhs - rhs] == 0
  ];
VerifyR3[expr_] := VerifyR3[1, expr];
```

```
Total[MapIndexed[(#1 /. f → f#2[[1]]) &, DeleteCases[FormalBasis[{j, k}, f], _β | _a]]]
```

$$c[f_5[b_j, b_k], j] + c[f_6[b_j, b_k], k] + na[f_1[b_j, b_k], j, j] + na[f_2[b_j, b_k], j, k] + na[f_3[b_j, b_k], k, j] + na[f_4[b_j, b_k], k, k] + nca[f_{11}[b_j, b_k], j, j, j] + nca[f_{12}[b_j, b_k], j, j, k] + nca[f_{13}[b_j, b_k], j, k, j] + nca[f_{14}[b_j, b_k], j, k, k] + nca[f_{15}[b_j, b_k], k, j, j] + nca[f_{16}[b_j, b_k], k, j, k] + nca[f_{17}[b_j, b_k], k, k, j] + nca[f_{18}[b_j, b_k], k, k, k] + n\delta a[f_7[b_j, b_k], j, j] + n\delta a[f_8[b_j, b_k], j, k] + n\delta a[f_9[b_j, b_k], k, j] + n\delta a[f_{10}[b_j, b_k], k, k] + n\delta aa[f_{19}[b_j, b_k], j, j, j, j] + n\delta aa[f_{20}[b_j, b_k], j, j, j, k] + n\delta aa[f_{21}[b_j, b_k], j, j, k, j] + n\delta aa[f_{22}[b_j, b_k], j, j, k, k] + n\delta aa[f_{23}[b_j, b_k], j, k, j, k] + n\delta aa[f_{24}[b_j, b_k], j, k, k, k] + n\delta aa[f_{25}[b_j, b_k], k, j, k, j] + n\delta aa[f_{26}[b_j, b_k], k, j, k, k] + n\delta aa[f_{27}[b_j, b_k], k, k, k, k]$$

```
f22[__] = 0; f21[__] = 0; f9[__] = 0; f5[__] = 0; f13[__] = 0; f17[__] = 0; f7[__] = 0;
f8[__] = 0;
f10[bj_, bk_] := g2[bk];
f1[bj_, bk_] := g3[bk];
f12[bj_, bk_] := -bj f19[bj, bk];
f19[bj_, bk_] := -(e^-bj (2 - 2 e^bj + bj + e^bj bj))/(2 bj^3);
f16[bj_, bk_] := g1[bj];
f11[bj_, bk_] := g4[bj];
f14[bj_, bk_] := -bj f20[bj, bk];
f20[bj_, bk_] := (-1 + e^bj)/(e^bj bj) (-2 + 2 e^bk - bk - e^bk bk + 2 bk^2 g2[bk] + 2 e^bk bk^2 g4[bk])/(2 (-1 + e^bk) bk^2);
f2[bj_, bk_] := g5[bj] - bj f4[bj, bk];
(* Non-forced choice: *) f4[__] = 0;
g5[bj_] := -(e^-bj (2 - 2 e^bj + bj + e^bj bj + 2 bj g3[bj]))/(2 bj);
(* Non-forced choices: *) g1[_] = 0; g2[_] = 0; g3[_] = 0;
g4[_] = 0; f3[__] = 0; f6[__] = 0; f15[__] = 0; f18[__] = 0; f23[__] = 0;
Total[MapIndexed[(#1 /. f → f#2[[1]]) &,
   DeleteCases[FormalBasis[{j, k}, f], _β | _na]]] // S
```

$$c\left[-\frac{e^{-b_j}\left(2-2e^{b_j}+\left(1+e^{b_j}\right)b_j\right)}{2b_j},\ k\right]+nca\left[\frac{e^{-b_j}\left(2-2e^{b_j}+\left(1+e^{b_j}\right)b_j\right)}{2b_j^2},\ k,\ j,\ k\right]+$$
$$nca\left[\frac{e^{-b_j}\left(-1+e^{b_j}\right)\left(2-2e^{b_k}+\left(1+e^{b_k}\right)b_k\right)}{2\left(-1+e^{b_k}\right)b_k^2},\ k,\ k,\ k\right]+$$
$$n\delta aa\left[-\frac{e^{-b_j}\left(2-2e^{b_j}+\left(1+e^{b_j}\right)b_j\right)}{2b_j^3},\ j,\ k,\ j,\ k\right]+$$
$$n\delta aa\left[-\frac{e^{-b_j}\left(-1+e^{b_j}\right)\left(2-2e^{b_k}+\left(1+e^{b_k}\right)b_k\right)}{2\left(-1+e^{b_k}\right)b_j b_k^2},\ j,\ k,\ k,\ k\right]$$

```
R[t_, j_, k_][x_] := Expand[
   x // Ad[na[t, j, k]] //
    (# + t B[Total[MapIndexed[(#1 /. f → f#2[[1]]) &, DeleteCases[
            FormalBasis[{j, k}, f], _β | _na]]], #]) &
  ];
R[j_, k_][x_] := R[1, j, k][x];

Print[VerifyR3[#]] & /@
  {na[f[b1, b2, b3, b4], 1, 4], na[f[b1, b2, b3, b4], 2, 4], na[f[b1, b2, b3, b4], 3, 4],
   na[f[b1, b2, b3, b4], 4, 1], na[f[b1, b2, b3, b4], 4, 2], na[f[b1, b2, b3, b4], 4, 3]};
```

$$na[f[b_1, b_2, b_3, b_4], 1, 4] \to \text{True}$$

$$na[f[b_1, b_2, b_3, b_4], 2, 4] \to \text{True}$$

$$na[f[b_1, b_2, b_3, b_4], 3, 4] \to \text{True}$$

$$na[f[b_1, b_2, b_3, b_4], 4, 1] \to \text{True}$$

$$na[f[b_1, b_2, b_3, b_4], 4, 2] \to \text{True}$$

$$na[f[b_1, b_2, b_3, b_4], 4, 3] \to \text{True}$$

```
Print[VerifyR3[ #]] & /@
  {na[f[b1, b2, b3, b4], 1, 2], na[f[b1, b2, b3, b4], 1, 3], na[f[b1, b2, b3, b4], 2, 3],
   na[f[b1, b2, b3, b4], 2, 1], na[f[b1, b2, b3, b4], 3, 1], na[f[b1, b2, b3, b4], 3, 2]};
```

$$na[f[b_1, b_2, b_3, b_4], 1, 2] \to \text{True}$$

$$na[f[b_1, b_2, b_3, b_4], 1, 3] \to \text{True}$$

$$na[f[b_1, b_2, b_3, b_4], 2, 3] \to \text{True}$$

$$na[f[b_1, b_2, b_3, b_4], 2, 1] \to \text{True}$$

$$na[f[b_1, b_2, b_3, b_4], 3, 1] \to \text{True}$$

$$na[f[b_1, b_2, b_3, b_4], 3, 2] \to \text{True}$$

```
Print[VerifyR3[t, #]] & /@
  {na[f[b1, b2, b3, b4], 1, 4], na[f[b1, b2, b3, b4], 2, 4], na[f[b1, b2, b3, b4], 3, 4],
   na[f[b1, b2, b3, b4], 4, 1], na[f[b1, b2, b3, b4], 4, 2], na[f[b1, b2, b3, b4], 4, 3]};
```

$$na[f[b_1, b_2, b_3, b_4], 1, 4] \to \text{True}$$

$$na[f[b_1, b_2, b_3, b_4], 2, 4] \to$$

$$\begin{aligned}
&nca\left[-\frac{1}{b_1}e^{-b_1-t\,b_2}\,(-1+e^{t\,b_2})\,f[b_1, b_2, b_3, b_4]\,(2\,(e^{b_1}+e^{(1+t)\,b_1}\,(-1+t)-e^{t\,b_1}\,t)+(e^{b_1}-e^{t\,b_1})\,t\,b_1),\right.\\
&\quad\left.3, 2, 4\right]+nca\left[\frac{1}{b_1^2}e^{-b_1-t\,b_2}\,(-1+e^{t\,b_2})\,f[b_1, b_2, b_3, b_4]\right.\\
&\quad\left.(2\,(e^{b_1}+e^{(1+t)\,b_1}\,(-1+t)-e^{t\,b_1}\,t)+(e^{b_1}-e^{t\,b_1})\,t\,b_1)\,b_2, 3, 1, 4\right]+\\
&n\delta aa\left[\frac{1}{b_1^2}e^{-b_1-t\,b_2}\,(-1+e^{t\,b_2})\,f[b_1, b_2, b_3, b_4]\,(2\,(e^{b_1}+e^{(1+t)\,b_1}\,(-1+t)-e^{t\,b_1}\,t)+(e^{b_1}-e^{t\,b_1})\,t\,b_1),\right.\\
&\quad\left.1, 3, 2, 4\right]+n\delta aa\left[-\frac{1}{b_1^3}e^{-b_1-t\,b_2}\,(-1+e^{t\,b_2})\,f[b_1, b_2, b_3, b_4]\right.\\
&\quad\left.(2\,(e^{b_1}+e^{(1+t)\,b_1}\,(-1+t)-e^{t\,b_1}\,t)+(e^{b_1}-e^{t\,b_1})\,t\,b_1)\,b_2, 1, 3, 1, 4\right]==0
\end{aligned}$$

$Aborted

```
ρ0 = Total[
  MapIndexed[(#1 /. f → f#2[[1]]) &, DeleteCases[FormalBasis[{j, k}, f], _β | _na]]]
```

$$c\left[-\frac{e^{-b_j}\left(2-2\,e^{b_j}+b_j+e^{b_j}\,b_j\right)}{2\,b_j},\,k\right]+nca\left[\frac{e^{-b_j}\left(2-2\,e^{b_j}+b_j+e^{b_j}\,b_j\right)}{2\,b_j^2},\,k,\,j,\,k\right]+$$
$$nca\left[-\frac{e^{-b_j}\left(-1+e^{b_j}\right)\left(-2+2\,e^{b_k}-b_k-e^{b_k}\,b_k\right)}{2\left(-1+e^{b_k}\right)b_k^2},\,k,\,k,\,k\right]+$$
$$n\delta aa\left[-\frac{e^{-b_j}\left(2-2\,e^{b_j}+b_j+e^{b_j}\,b_j\right)}{2\,b_j^3},\,j,\,k,\,j,\,k\right]+$$
$$n\delta aa\left[\frac{e^{-b_j}\left(-1+e^{b_j}\right)\left(-2+2\,e^{b_k}-b_k-e^{b_k}\,b_k\right)}{2\left(-1+e^{b_k}\right)b_j\,b_k^2},\,j,\,k,\,k,\,k\right]$$

```
ϕ1[x_] := e^-x - 1; ϕ2[x_] := ((x + 2) e^-x - 2 + x)/(2 x);
ρ = c[-ϕ2[bj], k] + nca[ϕ2[bj]/bj, k, j, k] + nca[ϕ1[bj]/(bk ϕ1[bk]) ϕ2[bk], k, k, k] +
    nδaa[-ϕ2[bj]/bj^2, j, k, j, k] + nδaa[-ϕ1[bj]/(bj bk ϕ1[bk]) ϕ2[bk], j, k, k, k];
ρ0 - ρ
```

```
0
```